Magnetic Field due to Helmholtz Coils

In[3]:=
```
Clear["Global`*"]
B[x_] := μ0 i R^2 / (2 (R^2 + x^2)^(3/2))
 (* Field due to a single coil of radius R a distance 'x' away *)
```

In[5]:=
```
Bleft[x_, s_] := B[x + s / 2]
Bright[x_, s_] := B[x - s / 2]
Btot[x_, s_] := Bleft[x, s] + Bright[x, s]
```

Evaluate Derivatives at the center

In[8]:=
```
Simplify[D[Btot[x, R], x]]
```

Out[8]=
$$48\, i\, R^2 \left(\frac{R - 2x}{2\left(5R^2 - 4Rx + 4x^2\right)^{5/2}} - \frac{\frac{R}{2} + x}{\left(5R^2 + 4Rx + 4x^2\right)^{5/2}} \right) \mu 0$$

In[9]:=
```
% /. x → 0
```

Out[9]=
```
0
```

In[10]:=
```
Simplify[D[Btot[x, R], {x, 2}]]
```

Out[10]=
$$768\, i\, R^2\, x \left(\frac{-R + x}{\left(5R^2 - 4Rx + 4x^2\right)^{7/2}} + \frac{R + x}{\left(5R^2 + 4Rx + 4x^2\right)^{7/2}} \right) \mu 0$$

In[11]:=
```
% /. x → 0
```

Out[11]=
```
0
```

In[12]:=
```
Simplify[D[Btot[x, R], {x, 3}]]
```

Out[12]=
$$3840\, i\, R^2 \left(\frac{R^3 - 6Rx^2 - 4x^3}{\left(5R^2 + 4Rx + 4x^2\right)^{9/2}} + \frac{-R^3 + 6Rx^2 - 4x^3}{\left(5R^2 - 4Rx + 4x^2\right)^{9/2}} \right) \mu 0$$

In[13]:=
```
% /. x → 0
```

Out[13]=
```
0
```

In[14]:=
```
Simplify[D[Btot[x, R], {x, 4}]]
```

Out[14]=
$$23\,040\, i\, R^2 \left(\frac{-3R^4 + 16R^3 x - 32Rx^3 + 16x^4}{\left(5R^2 - 4Rx + 4x^2\right)^{11/2}} + \frac{-3R^4 - 16R^3 x + 32Rx^3 + 16x^4}{\left(5R^2 + 4Rx + 4x^2\right)^{11/2}} \right) \mu 0$$

In[15]:=
```
% /. x → 0 n(* The 4th derivative is not zero at the center. *)
```

Out[15]=

$$-\frac{27\,648\,\mathrm{i}\,R^6\,\mu 0}{625\,\sqrt{5}\,\left(R^2\right)^{11/2}}$$

Plot the Field

Pick some plausible numbers.

In[16]:=
```
R = 0.15; μ0 = 4 π 10^-7; i = 5.0;
```

In[17]:=
```
Btot[x, s]
```

Out[17]=

$$\frac{7.06858\times 10^{-8}}{\left(0.0225+\left(-\frac{s}{2}+x\right)^2\right)^{3/2}}+\frac{7.06858\times 10^{-8}}{\left(0.0225+\left(\frac{s}{2}+x\right)^2\right)^{3/2}}$$

In[18]:=
```
BHC = Btot[0, R] (* Max value at center of Helmholtz Coil configuration *)
```

Out[18]=

```
0.0000299725
```

In[19]:=
```
Bmax = 2 * B[0] (* Max value for 2 overlapping coils *)
```

Out[19]=

```
0.0000418879
```

In[20]:=
```
Coil[x_, s_] :=
 Piecewise[{{Bmax / BHC, Abs[Abs[x] - s / 2] < 0.01}}, 0] (* Draw coils *)
```

In[21]:=
```
Manipulate[Plot[{Coil[x * R, s * R], Bleft[x * R, s * R] / BHC,
    Bright[x * R, s * R] / BHC, Btot[x * R, s * R] / BHC}, {x, -1.5, 1.5},
  PlotRange → {0, Bmax / BHC}, PlotPoints → 600, ImageSize → Scaled[0.9],
  LabelStyle → Larger,
  AxesLabel → {"x/R", "B / BHC"},
  PlotLegends → {"Coils", "Left", "Right", "Total"}],
 {{s, 2.0}, 0, 2.0, Appearance → "Labeled"}]
```

Out[21]=